Stage IREM de LILLE : "Calcul Formel"

A la recherche des formules BBP (Bailey, Borwein, Plouffe)

M.Gouy, G.Huvent, A.Ladureau

Calcul d'une intégrale

Pour calculer des intégrales du type $\int_0^1 \frac{\mathrm{P}(y)}{y^n - b}\, dy$, il faut avant tout transformer la fraction $\frac{\mathrm{P}(y)}{y^n - b}$ en éléments simples. Cette opération est effectuée par la commande suivante (sur un exemple)

```
> F:=convert((y^5+y^4+2*y^3-4)/(y^8-16),parfrac,y);
```

$$F := -\frac{1}{4}\frac{-2+y}{y^2-2\,y+2} + \frac{\frac{1}{4}\,y}{y^2-2}$$

Il faut ensuite calculer l'intégrale de chaque élément. Ceci est réalisé par la commande suivante

```
> map(f->Int(f,y=0..1),F);
```

$$\int_0^1 -\frac{1}{4}\frac{-2+y}{y^2-2\,y+2}\,dy + \int_0^1 \frac{1}{4}\frac{y}{y^2-2}\,dy$$

La commande map applique à chaque élément de la somme la fonction qui à f associe son intégrale de 0 à 1. Remarquez l'utilisation de Int qui est la forme inerte de l'intégrale. Comparez avec la commande suivante

```
> map(f->int(f,y=0..1),F);
```

$$\frac{1}{16}\,\pi$$

Il peut être intéressant de construire une fonction qui au polynôme P, et aux entier n et b associe le résultat final.

Conctruction de formule BBP par manipulation des expressions trouvées

Pour rechercher des formules BBP, voici, sur un exemple, la démarche adoptée.

On cherche des formules BBP du type $\sum_{i=0}^{\infty} \frac{1\left(\frac{a_0}{4\,i+1} + \frac{a_1}{4\,i+2} + \frac{a_2}{4\,i+3}\right)}{9^i}$

Pour cela on va calculer des intégrales du type $\displaystyle\int_0^1 \frac{a_0 + a_1\, y + a_2\, y^2}{y^4 - 9}\, dy$

On commence par créer un polynôme

```
> P:=add(a[i]*y^i,i=0..2);
```

$$P := a_0 + a_1\, y + a_2\, y^2$$

Puis la fraction rationnelle F, que l'on décompose ensuite en éléments simples

```
> F:=P/(y^4-9);
```

$$F := \frac{a_0 + a_1\, y + a_2\, y^2}{y^4 - 9}$$

```
> F:=convert(F,parfrac,y);
```

$$F := -\frac{1}{6}\,\frac{a_0 + a_1\, y - 3\, a_2}{y^2 + 3} + \frac{\frac{1}{6}\,(a_0 + a_1\, y + 3\, a_2)}{y^2 - 3}$$

Enfin, on calcule l'intégrale de F entre y=0 et y=1.

```
> resul:=map(f->int(f,y=0..1),F);
```

$$resul := -\frac{1}{12}\, a_1 \ln(2) - \frac{1}{108}\sqrt{3}\,\pi\, a_0 + \frac{1}{36}\sqrt{3}\,\pi\, a_2 - \frac{1}{18}\sqrt{3}\,\operatorname{arctanh}\left(\frac{1}{3}\sqrt{3}\right) a_0 - \frac{1}{6}\sqrt{3}\,\operatorname{arctanh}\left(\frac{1}{3}\sqrt{3}\right) a_2$$

Le résulat n'est pas très lisible, on regroupe les termes analogues.

```
> resul:=collect(resul,{ln(2),Pi,arctanh});
```

$$resul := \left(-\frac{1}{108}\sqrt{3}\; a_0 + \frac{1}{36}\sqrt{3}\; a_2\right)\pi - \frac{1}{12}\, a_1 \ln(2) + \left(-\frac{1}{18}\sqrt{3}\; a_0 - \frac{1}{6}\sqrt{3}\; a_2\right)\operatorname{arctanh}\left(\frac{1}{3}\sqrt{3}\right)$$

On cherche une formule pour $\pi\sqrt{3}$, on impose donc à a_1 d'être nul dans le polynôme P et dans le résultat.

```
> resul:=subs(a[1]=0,resul);
  P:=subs(a[1]=0,P);
```

$$resul := \left(-\frac{1}{108}\sqrt{3}\; a_0 + \frac{1}{36}\sqrt{3}\; a_2\right)\pi + \left(-\frac{1}{18}\sqrt{3}\; a_0 - \frac{1}{6}\sqrt{3}\; a_2\right)\operatorname{arctanh}\left(\frac{1}{3}\sqrt{3}\right)$$

$$P := a_0 + a_2\, y^2$$

Puis on élimine le terme en arctanh

```
> solve({-1/18*sqrt(3)*a[0]-1/6*sqrt(3)*a[2]});
```

$$\{a_0 = -3\, a_2,\ a_2 = a_2\}$$

```
> resul:=subs({a[2] = a[2], a[0] = -3*a[2]},resul);
  P:=subs({a[2] = a[2], a[0] = -3*a[2]},P);
```

$$resul := \frac{1}{18}\sqrt{3}\;\pi\, a_2$$

$$P := -3\, a_2 + a_2\, y^2$$

On peut enfin écrire la formule bbp correspondante

Ecriture des formules BBP

Soit $P(y) = a_0 + a_1\, y\ + ...$ un polynôme, n et b deux entiers naturels , la fonction bbp qui est donnée permet de transformer une intégrale en une formule du type BBP suivant l'égalité. Elle ne réalise aucun calcul mais permet un affichage plus agréable.

$$\int_0^1 \frac{\sum_{k=0}^{n-2} a_k\, y^k}{y^n - b}\, dy = \left(-\frac{1}{b}\right)\left(\sum_{i=0}^{\infty}\left(\sum_{k=0}^{n-1}\left[\frac{1}{b^i}\right]\left[\frac{a_k}{n\, i + k + 1}\right]\right)\right)$$

```
> bbp:=proc(n,b,P) global i; local S,somme,k;
  S:=1/b^i;
  somme:=0;
  for k from 0 to n-1 do
  somme :=somme +(coeff(P,y,k)/(n*i+k+1));
  od;
  -1/b*Sum(S*somme,i=0..infinity);
  end;
```

bbp := **proc**(n, b, P)

local *S*, *somme*, *k*;

global *i*;

S := 1 / b^i;

somme := 0;

for *k* **from** 0 **to** $n - 1$ **do** *somme* := *somme* + coeff(P, y, k) / (n*i + k + 1) **end do**;

−Sum(*somme***S*, $i = 0 .. \infty$) / b

end proc

Par exemple la formule de Plouffe donne

```
> 1/16*Pi=bbp(8,16,y^5+y^4+2*y^3-4);
```

$$\frac{1}{16}\pi = -\frac{1}{16}\left(\sum_{i=0}^{\infty}\frac{-4\frac{1}{8\,i+1}+\frac{2}{8\,i+4}+\frac{1}{8\,i+5}+\frac{1}{8\,i+6}}{16^i}\right)$$

Avec l'exemple précédent, on obtient

```
> formule_bbp:=bbp(4,9,P)=resul;
```

$$formule\_bbp := -\frac{1}{9}\left(\sum_{i=0}^{\infty} \frac{-3\,\frac{a_2}{4\,i+1}+\frac{a_2}{4\,i+3}}{9^i}\right)=\frac{1}{18}\sqrt{3}\,\pi\,a_2$$

```
> formule_bbp:=subs(a[2]=-1,formule_bbp);
```

$$formule\_bbp := -\frac{1}{9}\left(\sum_{i=0}^{\infty} \frac{3\,\frac{1}{4\,i+1}-\frac{1}{4\,i+3}}{9^i}\right)=-\frac{1}{18}\sqrt{3}\,\pi$$

```
> -4*formule_bbp;
```

$$\frac{4}{9}\left(\sum_{i=0}^{\infty} \frac{3\,\frac{1}{4\,i+1}-\frac{1}{4\,i+3}}{9^i}\right)=\frac{2}{9}\sqrt{3}\,\pi$$

```
> evalf(%,20);
```

$$1.2091995761561452337 = 1.2091995761561452337$$

```
>
>
>
```

Maintenant, à vous de trouver des formules BBP pour les constantes ln(3), ln(5),... en partant de l'intégrale $\displaystyle\int_0^1 \frac{P(y)}{y^8-16}\,dy$

Quelques éléments de réponse

On utilise la méthode indiquée (avec des améliorations de mise en page)

```
> restart:
  bbp:=proc(n,b,P) global i; local S,somme,k;
  S:=1/b^i;
  somme:=0;
  for k from 0 to n-1 do
  somme :=somme +(coeff(P,y,k)/(n*i+k+1));
  od;
  -1/b*Sum(S*somme,i=0..infinity);
  end;
```

bbp := **proc**(n, b, P)
local S, *somme*, k;
global i;
$S := 1 / b$^i;
somme := 0;
for k **from** 0 **to** $n-1$ **do** *somme* := *somme* + coeff(P, y, k) / ($n*i+k+1$) **end do**;
$-\mathrm{Sum}(S*somme, i = 0 .. \infty) / b$
end proc

```
> P:=add(a[i]*y^i,i=0..6);
  F:=P/(y^8-16);
  convert(F,parfrac,y);
```

$$P := a_0 + a_1\,y + a_2\,y^2 + a_3\,y^3 + a_4\,y^4 + a_5\,y^5 + a_6\,y^6$$

$$F := \frac{a_0 + a_1\,y + a_2\,y^2 + a_3\,y^3 + a_4\,y^4 + a_5\,y^5 + a_6\,y^6}{y^8 - 16}$$

$$\frac{1}{32}\,\frac{a_0 + 4\,a_4 + 8\,a_6 + 2\,a_2 + a_1\,y + 2\,y\,a_3 + 4\,y\,a_5}{y^2-2}$$
$$-\frac{1}{32}\,\frac{4\,a_4 - 8\,a_6 - 2\,a_2 + a_0 + a_1\,y - 2\,y\,a_3 + 4\,y\,a_5}{y^2+2}$$
$$-\frac{1}{64}\,\frac{-8\,a_4 + 2\,a_0 - 2\,a_1 + 4\,a_3 + 8\,a_5 + 4\,y\,a_3 - 4\,y\,a_4 + 8\,y\,a_6 - 2\,y\,a_2 + y\,a_0}{y^2+2\,y+2}$$

$$+\frac{\frac{1}{64}(-2\,a_1+4\,a_3+8\,a_5+8\,a_4-2\,a_0-4\,y\,a_3-4\,y\,a_4+8\,y\,a_6-2\,y\,a_2+y\,a_0)}{y^2-2\,y+2}$$

```
> resul:=bbp(8,16,P)=map(f->int(f,y=0..1),%);
```

$$resul := -\frac{1}{16}\left(\sum_{i=0}^{\infty}\frac{\frac{a_0}{8\,i+1}+\frac{a_1}{8\,i+2}+\frac{a_2}{8\,i+3}+\frac{a_3}{8\,i+4}+\frac{a_4}{8\,i+5}+\frac{a_5}{8\,i+6}+\frac{a_6}{8\,i+7}}{16^i}\right)=$$
$$-\frac{1}{16}\sqrt{2}\,\mathrm{arctan}\left(\frac{1}{2}\sqrt{2}\right)a_4-\frac{1}{64}\sqrt{2}\,\mathrm{arctanh}\left(\frac{1}{2}\sqrt{2}\right)a_0-\frac{1}{16}\sqrt{2}\,\mathrm{arctanh}\left(\frac{1}{2}\sqrt{2}\right)a_4$$
$$-\frac{1}{32}\ln(5)\,a_3+\frac{1}{32}\ln(5)\,a_4-\frac{1}{16}\ln(5)\,a_6+\frac{1}{64}\ln(5)\,a_2-\frac{1}{128}\ln(5)\,a_0+\frac{1}{32}\mathrm{arctan}(2)\,a_1$$
$$-\frac{1}{8}\mathrm{arctan}(2)\,a_5+\frac{1}{16}\mathrm{arctan}(2)\,a_4-\frac{1}{64}\mathrm{arctan}(2)\,a_0+\frac{1}{8}\mathrm{arctan}(2)\,a_6-\frac{1}{32}\mathrm{arctan}(2)\,a_2$$
$$-\frac{1}{32}\sqrt{2}\,\mathrm{arctanh}\left(\frac{1}{2}\sqrt{2}\right)a_2-\frac{1}{64}\ln(3)\,a_1+\frac{1}{32}\ln(3)\,a_3+\frac{1}{32}\sqrt{2}\,\mathrm{arctan}\left(\frac{1}{2}\sqrt{2}\right)a_2$$
$$-\frac{1}{64}\sqrt{2}\,\mathrm{arctan}\left(\frac{1}{2}\sqrt{2}\right)a_0-\frac{1}{64}\pi\,a_1+\frac{1}{16}\pi\,a_5-\frac{1}{16}\ln(3)\,a_5-\frac{1}{8}\sqrt{2}\,\mathrm{arctanh}\left(\frac{1}{2}\sqrt{2}\right)a_6$$
$$+\frac{1}{8}\sqrt{2}\,\mathrm{arctan}\left(\frac{1}{2}\sqrt{2}\right)a_6$$

```
> resul:=collect(%,{Pi,ln,arctan,arctanh});
```

$$resul := -\frac{1}{16}\left(\sum_{i=0}^{\infty}\frac{\frac{a_0}{8\,i+1}+\frac{a_1}{8\,i+2}+\frac{a_2}{8\,i+3}+\frac{a_3}{8\,i+4}+\frac{a_4}{8\,i+5}+\frac{a_5}{8\,i+6}+\frac{a_6}{8\,i+7}}{16^i}\right)=$$
$$\left(-\frac{1}{32}\sqrt{2}\,a_2-\frac{1}{64}\sqrt{2}\,a_0-\frac{1}{8}\sqrt{2}\,a_6-\frac{1}{16}\sqrt{2}\,a_4\right)\mathrm{arctanh}\left(\frac{1}{2}\sqrt{2}\right)$$
$$+\left(\frac{1}{32}a_1+\frac{1}{16}a_4+\frac{1}{8}a_6-\frac{1}{32}a_2-\frac{1}{8}a_5-\frac{1}{64}a_0\right)\mathrm{arctan}(2)$$
$$+\left(\frac{1}{8}\sqrt{2}\,a_6+\frac{1}{32}\sqrt{2}\,a_2-\frac{1}{64}\sqrt{2}\,a_0-\frac{1}{16}\sqrt{2}\,a_4\right)\mathrm{arctan}\left(\frac{1}{2}\sqrt{2}\right)$$
$$+\left(-\frac{1}{16}a_6-\frac{1}{128}a_0-\frac{1}{32}a_3+\frac{1}{32}a_4+\frac{1}{64}a_2\right)\ln(5)+\left(-\frac{1}{16}a_5-\frac{1}{64}a_1+\frac{1}{32}a_3\right)\ln(3)$$
$$+\left(\frac{1}{16}a_5-\frac{1}{64}a_1\right)\pi$$

Formules pour π

```
> solve({-1/64*a[1]+1/32*a[3]-1/16*a[5],-1/16*a[6]+1/64*a[2]-1/32*
a[3]+1/32*a[4]-1/128*a[0],1/32*a[1]-1/32*a[2]+1/16*a[4]-1/64*a[0
]-1/8*a[5]+1/8*a[6],-1/64*sqrt(2)*a[0]-1/16*sqrt(2)*a[4]+1/32*sq
rt(2)*a[2]+1/8*sqrt(2)*a[6],-1/64*sqrt(2)*a[0]-1/32*sqrt(2)*a[2]
```

```
-1/16*sqrt(2)*a[4]-1/8*sqrt(2)*a[6]});
```

$$\{a_1 = -8\,a_6,\ a_6 = a_6,\ a_3 = a_3,\ a_5 = \frac{1}{2}a_3 + 2\,a_6,\ a_0 = -2\,a_3 - 8\,a_6,\ a_2 = -4\,a_6,\ a_4 = \frac{1}{2}a_3 + 2\,a_6\}$$

```
> resul_Pi:=subs(%,resul);
```

$$resul\_Pi := -\frac{1}{16}\left(\sum_{i=0}^{\infty}\frac{\frac{-2\,a_3-8\,a_6}{8\,i+1}-\frac{8\,a_6}{8\,i+2}-\frac{4\,a_6}{8\,i+3}+\frac{a_3}{8\,i+4}+\frac{\frac{1}{2}a_3+2\,a_6}{8\,i+5}+\frac{\frac{1}{2}a_3+2\,a_6}{8\,i+6}+\frac{a_6}{8\,i+7}}{16^i}\right)=$$
$$\left(-\frac{1}{64}\sqrt{2}\,(-2\,a_3-8\,a_6)-\frac{1}{16}\sqrt{2}\left(\frac{1}{2}a_3+2\,a_6\right)\right)\operatorname{arctanh}\left(\frac{1}{2}\sqrt{2}\right)$$
$$+\left(-\frac{1}{64}\sqrt{2}\,(-2\,a_3-8\,a_6)-\frac{1}{16}\sqrt{2}\left(\frac{1}{2}a_3+2\,a_6\right)\right)\arctan\left(\frac{1}{2}\sqrt{2}\right)+\left(\frac{1}{32}a_3+\frac{1}{4}a_6\right)\pi$$

Ce qui donne la formule d'Adamchik-Wagon avec

```
> -16*subs(a[5]=-2*a[6]-1,resul_Pi);
```

$$\sum_{i=0}^{\infty}\frac{\frac{-2\,a_3-8\,a_6}{8\,i+1}-\frac{8\,a_6}{8\,i+2}-\frac{4\,a_6}{8\,i+3}+\frac{a_3}{8\,i+4}+\frac{\frac{1}{2}a_3+2\,a_6}{8\,i+5}+\frac{\frac{1}{2}a_3+2\,a_6}{8\,i+6}+\frac{a_6}{8\,i+7}}{16^i}=$$
$$-16\left(-\frac{1}{64}\sqrt{2}\,(-2\,a_3-8\,a_6)-\frac{1}{16}\sqrt{2}\left(\frac{1}{2}a_3+2\,a_6\right)\right)\operatorname{arctanh}\left(\frac{1}{2}\sqrt{2}\right)$$
$$-16\left(-\frac{1}{64}\sqrt{2}\,(-2\,a_3-8\,a_6)-\frac{1}{16}\sqrt{2}\left(\frac{1}{2}a_3+2\,a_6\right)\right)\arctan\left(\frac{1}{2}\sqrt{2}\right)-16\left(\frac{1}{32}a_3+\frac{1}{4}a_6\right)\pi$$

Pour obtenir des formules pour ln(3) on peut procéder ainsi

```
> resul_ln3:=subs(solve({-1/16*a[6]+1/64*a[2]-1/32*a[3]+1/32*a[4]-
  1/128*a[0],-1/64*a[1]+1/16*a[5],1/32*a[1]-1/32*a[2]+1/16*a[4]-1/
  64*a[0]-1/8*a[5]+1/8*a[6],-1/64*sqrt(2)*a[0]-1/16*sqrt(2)*a[4]+1
  /32*sqrt(2)*a[2]+1/8*sqrt(2)*a[6],-1/64*sqrt(2)*a[0]-1/32*sqrt(2
  )*a[2]-1/16*sqrt(2)*a[4]-1/8*sqrt(2)*a[6]}),resul);
```

$$resul\_ln3 := -\frac{1}{16}\left(\sum_{i=0}^{\infty}\frac{8\frac{a_6}{8\,i+1}+\frac{a_1}{8\,i+2}-\frac{4\,a_6}{8\,i+3}-\frac{8\,a_6}{8\,i+4}-\frac{2\,a_6}{8\,i+5}+\frac{\frac{1}{4}a_1}{8\,i+6}+\frac{a_6}{8\,i+7}}{16^i}\right)=$$
$$\left(-\frac{1}{32}a_1-\frac{1}{4}a_6\right)\ln(3)$$

et particulariser ensuite.

Une remarque intéressante:

On peut également écrire 0 comme résultat d'une formule BBP (Ce qui n'a pas, a priori, grand intêret pour le calcul des chiffres binaires de 0). Par exemple

```
> resul;
```

$$-\frac{1}{16}\left(\sum_{i=0}^{\infty}\frac{\frac{a_0}{8i+1}+\frac{a_1}{8i+2}+\frac{a_2}{8i+3}+\frac{a_3}{8i+4}+\frac{a_4}{8i+5}+\frac{a_5}{8i+6}+\frac{a_6}{8i+7}}{16^i}\right)=$$

$$\left(-\frac{1}{32}\sqrt{2}\,a_2-\frac{1}{64}\sqrt{2}\,a_0-\frac{1}{8}\sqrt{2}\,a_6-\frac{1}{16}\sqrt{2}\,a_4\right)\operatorname{arctanh}\left(\frac{1}{2}\sqrt{2}\right)$$
$$+\left(\frac{1}{32}a_1+\frac{1}{16}a_4+\frac{1}{8}a_6-\frac{1}{32}a_2-\frac{1}{8}a_5-\frac{1}{64}a_0\right)\arctan(2)$$
$$+\left(\frac{1}{8}\sqrt{2}\,a_6+\frac{1}{32}\sqrt{2}\,a_2-\frac{1}{64}\sqrt{2}\,a_0-\frac{1}{16}\sqrt{2}\,a_4\right)\arctan\left(\frac{1}{2}\sqrt{2}\right)$$
$$+\left(-\frac{1}{16}a_6-\frac{1}{128}a_0-\frac{1}{32}a_3+\frac{1}{32}a_4+\frac{1}{64}a_2\right)\ln(5)+\left(-\frac{1}{16}a_5-\frac{1}{64}a_1+\frac{1}{32}a_3\right)\ln(3)$$
$$+\left(\frac{1}{16}a_5-\frac{1}{64}a_1\right)\pi$$

```
> resul_0:=subs(solve({-1/32*sqrt(2)*a[2]-1/64*sqrt(2)*a[0]-1/8*sqrt(2)*a[6]-1/16*sqrt(2)*a[4],1/32*a[1]+1/16*a[4]+1/8*a[6]-1/32*a[2]-1/8*a[5]-1/64*a[0],1/8*sqrt(2)*a[6]+1/32*sqrt(2)*a[2]-1/64*sqrt(2)*a[0]-1/16*sqrt(2)*a[4],-1/16*a[6]-1/128*a[0]-1/32*a[3]+1/32*a[4]+1/64*a[2],-1/16*a[5]-1/64*a[1]+1/32*a[3],1/16*a[5]-1/64*a[1]}),resul);
```

$$resul\_0 := -\frac{1}{16}\left(\sum_{i=0}^{\infty}\frac{8\frac{a_6}{8i+1}-\frac{8a_6}{8i+2}-\frac{4a_6}{8i+3}-\frac{8a_6}{8i+4}-\frac{2a_6}{8i+5}-\frac{2a_6}{8i+6}+\frac{a_6}{8i+7}}{16^i}\right)=0$$

Dans cette égalité on élève les dénominateurs au carré et on fait $a_6 = 1$

```
> resul_0_carr :=
subs(a[6]=1,-1/16*Sum(1/(16^i)*(8*a[6]/(8*i+1)^2-8*a[6]/(8*i+2)^2-4*a[6]/(8*i+3)^2-8*a[6]/(8*i+4)^2-2*a[6]/(8*i+5)^2-2*a[6]/(8*i+6)^2+a[6]/(8*i+7)^2),i = 0 .. infinity));
```

$$resul\_0\_carr := -\frac{1}{16}\left(\sum_{i=0}^{\infty}\frac{8\frac{1}{(8i+1)^2}-\frac{8}{(8i+2)^2}-\frac{4}{(8i+3)^2}-\frac{8}{(8i+4)^2}-\frac{2}{(8i+5)^2}-\frac{2}{(8i+6)^2}+\frac{1}{(8i+7)^2}}{16^i}\right)$$

```
>
```

si on demande une valeur approchée de cette somme

```
> evalf(resul_0_carr,20);
```

$$-.30842513753404245684$$

Divisons par $\frac{\pi^2}{6} = \zeta(2)$ pour obtenir

```
> evalf(resul_0_carr/Zeta(2));
```

$$-.1875000000$$

Magique ! On a numériquement (et on peut le prouver)

```
> %*16;
```

$$-3.000000000$$

```
>
> Sum(1/(16^i)*(8*1/((8*i+1)^2)-8/(8*i+2)^2-4/(8*i+3)^2-8/(8*i+4)^
  2-2/(8*i+5)^2-2/(8*i+6)^2+1/((8*i+7)^2)),i = 0 ..
  infinity)=3*Zeta(2);
```

$$\sum_{i=0}^{\infty} \frac{8\frac{1}{(8\,i+1)^2} - \frac{8}{(8\,i+2)^2} - \frac{4}{(8\,i+3)^2} - \frac{8}{(8\,i+4)^2} - \frac{2}{(8\,i+5)^2} - \frac{2}{(8\,i+6)^2} + \frac{1}{(8\,i+7)^2}}{16^i}$$

$$= \frac{1}{2}\pi^2$$

Avec cette heuristique, j'ai trouvé (puis démontré) des formules BBP jusqu'a $\zeta(5)$.

```
>
```

Maintenant à vous de chercher !